Zadání: Vyjádřete analytický předpis funkce průhybové čáry nosníku zatíženého a podepřeného dle obrázku.



Při vyšetřování průhybu nosníku analytickým způsobem vycházíme z (Bernoulliho) differenciální rovnice průhybové čáry

$$\mathrm{w}''(x) = -\frac{M(x)}{EI_y}.$$

Složitost výpočtu je dána zvoleným souřadným systémem, ale výsledný tvar deformovaného nosníku musí být stejný.

Postup výpočtu od volného konce

$$\begin{aligned}
\overleftarrow{x_1} \in \langle 0, L)\ \ M(x_1) &= -qx_1\frac{x_1}{2} = -\frac{1}{2}qx_1^2 \\
\overleftarrow{x_1} \in \langle L, 2L)\ \ M(x_1) &= -qL\left(x_1 - \frac{L}{2}\right) = -qLx_1 + \frac{1}{2}qL^2
\end{aligned}$$

```
M1:-(1/2)*q*x^2; /* M(x) in <0;L) */
M2:-q*L*(x-L/2); /* M(x) in <L;2L) */
expand(M2); /* polynom */
```

$$\begin{aligned}
&\overleftarrow{x_1} \in \langle L, 2L) \\
EI_y\varphi(x_1) &= \frac{1}{2}qx_1^2 - \frac{1}{2}qL^2x_1 + C_1 \\
\varphi(2L) = 0 &= \frac{1}{2}q(2L)^2 - \frac{1}{2}qL^2 2L + C_1 \\
C_1 &= -qL^3 \\
EI_y\varphi(x_1) &= \frac{1}{2}qx_1^2 - \frac{1}{2}qL^2x_1 - qL^3
\end{aligned}$$

```
phi2:integrate(-M2,x); /* phi(x) in <L;2L) */
C1:-subst(2*L,x,phi2); /* B.C. phi(2L) = 0 */
phi2:phi2+C1;
expand(phi2); /* final phi(x) function (polynom) */
```

$$\begin{aligned}
EI_y\mathrm{w}(x_1) &= \frac{1}{6}qx_1^3 - \frac{1}{4}qL^2x_1^2 - qL^3x_1 + C_2 \\
\mathrm{w}(2L) = 0 &= \frac{1}{6}q(2L)^3 - \frac{1}{4}qL^2(2L)^2 - qL^3 2L + C_2 \\
C_2 &= \frac{5}{3}qL^4 \\
EI_y\mathrm{w}(x_1) &= \frac{1}{6}qx_1^3 - \frac{1}{4}qL^2x_1^2 - qL^3x_1 + \frac{5}{3}qL^4
\end{aligned}$$

```
w2:integrate(phi2,x); /* w(x) in <L;2L) */
C2:-subst(2*L,x,w2); /* B.C. w(2L) = 0 */
w2:w2+C2;
expand(w2); /* final w(x) function (polynom) */
```

Pootoočení a průhyb nosníku v místě $x_1 = L$

$$\begin{aligned} EI_y\varphi(L) &= -qL^3 \\ EI_y\mathrm{w}(L) &= \frac{7}{12}qL^4 \end{aligned}$$

```
phil:subst(L,x,phi2); /* phi(L) */
wl:subst(L,x,w2); /* w(L) */
```

$$\begin{aligned} \overleftarrow{x_1} \in \langle 0, L) & \\ EI_y\varphi(x_1) &= \frac{1}{6}qx_1^3 + C_3 \\ EI_y\varphi(L) = -qL^3 &= \frac{1}{6}qL^3 + C_3 \\ C_3 &= -\frac{7}{6}qL^3 \\ EI_y\varphi(x_1) &= \frac{1}{6}qx_1^3 - \frac{7}{6}qL^3 \end{aligned}$$

```
phi1:integrate(-M1,x); /* phi(x) in <0;L) */
C3:-subst(L,x,phi1)+phil; /* B.C. phi1(L) =  phi2(L) */
phi1:phi1+C3;
expand(phi1); /* final phi(x) function (polynom) */
```

$$\begin{aligned} EI_y\mathrm{w}(x_1) &= \frac{1}{24}qx_1^4 - \frac{7}{6}qL^3x_1 + C_4 \\ EI_y\mathrm{w}(L) = \frac{7}{12}qL^4 &= \frac{1}{24}qL^4 - \frac{7}{6}qL^4 + C_4 \\ C_4 &= \frac{41}{24}qL^4 \\ EI_y\mathrm{w}(x_1) &= \frac{1}{24}qx_1^4 - \frac{7}{6}qL^3x_1 + \frac{41}{24}qL^4 \end{aligned}$$

```
w1:integrate(phi1,x); /* w(x) in <0;L) */
C4:-subst(L,x,w1)+wl; /* B.C. w1(L) =  w2(L) */
w1:w1+C4;
expand(w1); /* final w(x) function (polynom) */
```

Maximální průhyb nosníku

$$EI_y\mathrm{w}(0) = \frac{41}{24}qL^4$$

```
wmax:subst(0,x,w1); /* max deflection at w(0) */
```

Postup výpočtu od vetknutí

$$\begin{aligned} A &= qL \\ M_\mathrm{a} &= qL\frac{3}{2}L = \frac{3}{2}qL^2 \\ \overrightarrow{x_2} \in \langle 0, L) \;\; M(x_2) &= Ax_2 - M_\mathrm{a} = qLx_2 - \frac{3}{2}qL^2 \\ \overrightarrow{x_2} \in \langle L, 2L) \;\; M(x_2) &= Ax_2 - M_\mathrm{a} - q(x_2 - L)\frac{x_2 - L}{2} = -\frac{1}{2}qx_2^2 + 2qLx_2 - 2qL^2 \end{aligned}$$

```
A:q*L; /* vertical reaction */
Ma:q*L*(3/2)*L; /* moment reaction */
M1:A*x-Ma; /* M(x) in <0;L) */
M2:A*x-Ma-(1/2)*q*(x-L)^2; /* M(x) in <L;2L) */
expand(M2); /* polynom */
Ctrl1:subst(2*L,x,M2); /* zero bending moment at free end */
```

$$
\begin{aligned}
&\overrightarrow{x_2} \in \langle 0, L) \\
EI_y\varphi(x_2) &= -\frac{1}{2}qLx_2^2 + \frac{3}{2}qL^2x_2 + D_1 \\
\varphi(0) = 0 &= D_1 \\
EI_y\varphi(x_2) &= -\frac{1}{2}qLx_2^2 + \frac{3}{2}qL^2x_2
\end{aligned}
$$

```
phi1:integrate(-M1,x); /* phi(x) in <0;L) */
D1:-subst(0,x,phi1); /* B.C. phi(0) = 0 */
phi1:phi1+D1; /* final phi(x) function (polynom) */
```

$$
\begin{aligned}
EI_y\mathrm{w}(x_2) &= -\frac{1}{6}qLx_2^3 + \frac{3}{4}qL^2x_2^2 + D_1 \\
\mathrm{w}(0) = 0 &= D_2 \\
EI_y\mathrm{w}(x_2) &= -\frac{1}{6}qLx_2^3 + \frac{3}{4}qL^2x_2^2
\end{aligned}
$$

```
w1:integrate(phi1,x); /* w(x) in <0;L) */
D2:subst(0,x,w1); /* B.C. w(0) = 0 */
w1:w1+D2; /* final w(x) function (polynom) */
```

Pootočení a průhyb nosníku v místě $x_2 = L$

$$
\begin{aligned}
EI_y\varphi(L) &= qL^3 \\
EI_y\mathrm{w}(L) &= \frac{7}{12}qL^4
\end{aligned}
$$

```
phil:subst(L,x,phi1); /* phi(L) */
wl:subst(L,x,w1); /* w(L) */
```

$$
\begin{aligned}
&\overrightarrow{x_2} \in \langle L, 2L) \\
EI_y\varphi(x_2) &= \frac{1}{6}qx_2^3 - qLx_2^2 + 2qL^2x_2 + D_3 \\
\varphi(L) = qL^3 &= \frac{1}{6}qL^3 - qL^3 + 2qL^3 + D_3 \\
D_3 &= -\frac{1}{6}qL^3 \\
EI_y\varphi(x_2) &= \frac{1}{6}qx_2^3 - qLx_2^2 + 2qL^2x_2 - \frac{1}{6}qL^3
\end{aligned}
$$

```
phi2:integrate(-M2,x); /* phi(x) in <L;2L) */
D3:-subst(L,x,phi2)+phil; /* B.C. phi1(L) =  phi2(L) */
phi2:phi2+D3;
expand(phi2); /* final phi(x) function (polynom) */
```

$$
\begin{aligned}
EI_y \mathrm{w}(x_2) &= \frac{1}{24}qx_2^4 - \frac{1}{3}qLx_2^3 + qL^2x_2^2 - \frac{1}{6}qL^3x_2 + D_4 \\
\mathrm{w}(L) = \frac{7}{12}qL^4 &= \frac{1}{24}qL^4 - \frac{1}{3}qL^4 + qL^4 - \frac{1}{6}qL^4 + D_4 \\
D_4 &= \frac{1}{24}qL^4 \\
EI_y \mathrm{w}(x_2) &= \frac{1}{24}qx_2^4 - \frac{1}{3}qLx_2^3 + qL^2x_2^2 - \frac{1}{6}qL^3x_2 + \frac{1}{24}qL^4
\end{aligned}
$$

```
w2:integrate(phi2,x); /* w(x) in <L;2L) */
D4:-subst(L,x,w2)+wl; /* B.C. w1(L) =  w2(L) */
w2:w2+D4;
expand(w2); /* final w(x) function (polynom) */
```

Maximální průhyb nosníku

$$
EI_y \mathrm{w}(2L) = \frac{41}{24}qL^4
$$

```
wmax:subst(2*L,x,w2); /* max deflection at w(2L) */
```





Výše uvedené postupy ukazují nejrychlejší řešení (bez využití SW nástrojů) od intervalu $\overrightarrow{x_2} \in \langle 0, L)$ pro vyjádření $\varphi(L)$, $\mathrm{w}(L)$ a následně $\overleftarrow{x_1} \in \langle 0, L)$.